
PLS205 Winter 2015 

 

Homework Topics 6-7 
 

Due Thursday, February 5, at the beginning of lab.  Answer all parts of the questions completely, and 

clearly document the procedures used in each exercise.  To ensure maximum points for yourself, invest 

some time in presenting your answers in a concise, organized, and clear manner. 

 

Question 1 [40 points] 

 

Recall question 1 from homework 4-5 to measure the yield of a molecular compound extraction. The 

graduate student wanted to repeat the experiment but decided to use only six treatments by 

combining three different tissue amounts (50g, 100 g, 150 g) and the presence or absence of the 

supplement (Notice that the 200g treatment has been dropped). In homework 4-5 the student did all 

the extractions by himself. In the current experiment the graduate student used the help of four 

undergraduate students in the lab to do the experiment, and wants to use a statistical analysis that 

will control for potential variability generated by the different students. Each undergraduate student 

tested all 6 treatments at the same time (one of each supplement and tissue combination) and 

independently of the other students. The results of the experiment are reported in the table below: 
  

Treatment 1 2 3 4 5 6 

 

  Supplement   No Supplement 

 

50 g 100 g 150 g 50 g 100 g 150 g 

undergrad 1 116.10 430.60 616.80 115.30 463.50 424.40 

undergrad 2 116.85 424.65 606.15 98.75 453.55 403.85 

undergrad 3 75.15 420.65 573.55 87.55 451.75 390.05 

undergrad 4 92.60 425.20 587.40 109.00 450.60 403.90 

 
Data hw67_1; 

Input undergrad $ trtmt $ yield; 

cards; 

ug1 trtmt1 116.1 

ug2 trtmt1 116.85 

ug3 trtmt1 75.15 

ug4 trtmt1 92.6 

ug1 trtmt2 430.6 

ug2 trtmt2 424.65 

ug3 trtmt2 420.65 

ug4 trtmt2 425.2 

ug1 trtmt3 616.8 

ug2 trtmt3 606.15 

ug3 trtmt3 573.55 

ug4 trtmt3 587.4 

ug1 trtmt4 115.3 

ug2 trtmt4 98.75 

ug3 trtmt4 87.55 

ug4 trtmt4 109 

ug1 trtmt5 463.5 

ug2 trtmt5 453.55 



ug3 trtmt5 451.75 

ug4 trtmt5 450.6 

ug1 trtmt6 424.4 

ug2 trtmt6 403.85 

ug3 trtmt6 390.05 

ug4 trtmt6 403.9 

; 

Proc Print data = hw67_1; 

Proc GLM data = hw67_1 Order = Data; 

Title 'RCBD'; 

Class undergrad trtmt; 

Model yield = undergrad trtmt; 

Output Out = hw67_1pr r = res p = pred; 

 Means trtmt undergrad; 

  contrast 'supplement vs. no-supplement' trtmt  1  1 1 -1 -1 -1; 

  contrast 'lineal'                    trtmt -1  0 1 -1  0  1; 

  contrast 'quadratic'                    trtmt  1 -2 1  1 -2  1; 

  contrast 'linear interaction'           trtmt -1  0 1  1  0 -1; 

  contrast 'quad interaction'             trtmt  1 -2 1 -1  2 -1; 

 

Proc GLM data = hw67_1pr; 

title 'Levene'; 

Class trtmt; 

Model yield = trtmt; 

Means trtmt / HovTest = Levene; 

 

Proc GLM data = hw67_1pr; 

title '1-df additivity test'; 

Class undergrad trtmt; 

Model yield = undergrad trtmt pred*pred; 

 

Proc Univariate Normal Plot Data = hw67_1pr; 

Title 'Normality'; 

var res; 

 

Proc Gplot data = hw67_1pr; 

title 'Res vs. Pred plot'; 

Plot res*pred = trtmt; 

 

Proc GLMPower Data = hw67_1; 

title 'power of RCBD'; 

Class undergrad trtmt; 

Model yield = undergrad trtmt; 

Power 

 stddev = 8.450738 

 alpha = 0.05 

 ntotal = 24 

 power = .; 

 

Proc GLM data = hw67_1; 

Title 'CRD'; 

Class trtmt; 

Model yield = trtmt; 

Means trtmt; 

 

Proc Power; 

title 'power of CRD'; 



onewayanova test = overall_f 

groupmeans = 100.175 | 425.275 | 595.975 | 102.65 | 454.85 | 405.55 

stddev = 13.99999 

npergroup = 4 

alpha = 0.05 

power = .; 

run; quit; 

 

1.1 Describe the design of the experiment using the table in the appendix. 

Design:  RCBD 

Response Variable:  Yield of molecular compound 

Experimental Unit:  extraction 

    

Class 

Variable 

Block or 

Treatment 

No. of 

Levels 
Description 

1 Block  4   Undergraduate student 

2 Treatment  8   Extraction supplement and tissue amount combinations 

↓       

n       

    

  Subsamples?  NO   

 

1.2 Test the assumptions of normality of the residuals, homogeneity of variances, additivity of 

main effects, and construct a plot of the residuals vs. the predicted values. Are there any patterns 

in the residuals by predicted plot? 

Homogeneity of variances: 

Levene's Test for Homogeneity of yield Variance 

ANOVA of Squared Deviations from Group Means 

Source DF Sum of Squares Mean Square F Value Pr > F 

trtmt 5 303544 60708.9 2.47 0.0718 

Error 18 442773 24598.5     

The Levene test for homogeneity of variances suggest that the variances are equal between 

the treatments (p = 0.0718). 

 

Shapiro - Wilk Test for Normality: 

Tests for Normality 

Test Statistic p Value 

Shapiro-Wilk W 0.951068 Pr < W 0.2857 

The Shapiro – Wilk test for normality suggest that the residuals fit a normal distribution (p = 

0.2857). 

 

Tukey 1 – df Test: 

Source DF Type III SS Mean Square F Value Pr > F 



Source DF Type III SS Mean Square F Value Pr > F 

undergrad 3 671.85331 223.95110 2.95 0.0691 

trtmt 5 35618.38517 7123.67703 93.90 <.0001 

pred*pred 1 9.11051 9.11051 0.12 0.7341 

The 1 – df test suggest that there are no significant block by treatment interactions (p = 

0.7341). 

 

 
 

The residual by predicted plot does not demonstrate any obvious patterns. 

 

1.3 Perform the correct statistical analysis to test if there are differences between the treatments 

and report the results of the ANOVA and your conclusions. Include the contrast to answer the 

following questions: 

 

Source DF Type III SS Mean Square F Value Pr > F 

undergrad 3 2456.7679 818.9226 11.47 0.0004 

trtmt 5 815214.9338 163042.9868 2283.04 <.0001 

 



The ANOVA results suggest that there are significant differences between treatments with 

different tissue amounts and addition of supplement (p < 0.0001). The  ANOVA suggest that 

there were significant blocking effects (p < 0.0004). Some undergraduates were better at 

extracting the molecular compound than others. 

 

a. Test if there is a difference between using a supplement and not using a supplement. 

Contrast DF Contrast SS Mean Square F Value Pr > F 

supplement vs. no-supplement 1 16721.7604 16721.7604 234.15 <.0001 

The 1 – df test suggest that there are significant differences in yield between adding a 

supplement and not adding the supplement (p < 0.0001). 

 

b. Test if there is a linear trend of yield by increasing tissue amount 

Contrast DF Contrast SS Mean Square F Value Pr > F 

lineal 1 637921.6900 637921.6900 8932.60 <.0001 

The 1 – df test suggest that there are significant linear trend of yield by tissue amount (p 

< 0.0001). 

 

c. Test if there is a quadratic trend of yield by increasing tissue amount 

Contrast DF Contrast SS Mean Square F Value Pr > F 

quadratic 1 103008.2700 103008.2700 1442.39 <.0001 

The 1 – df test suggest that there is a significant quadratic trend of yield by increasing 

tissue amount (p < 0.0001). 

 

d. Test if the linear trend is different in the presence of supplement vs. the absence of 

supplement. 

Contrast DF Contrast SS Mean Square F Value Pr > F 

linear interaction 1 37210.4100 37210.4100 521.04 <.0001 

The 1 – df test suggest that the linear trend in yield by increasing tissue amount is 

different when a supplement applied compared to when no supplement is applied (p < 

0.0001). 

 

e. Test if the quadratic trend is different in the presence of supplement vs. the absence of 

supplement 

Contrast DF Contrast SS Mean Square F Value Pr > F 

quad interaction 1 20352.8033 20352.8033 284.99 <.0001 

The 1 – df test suggest that the quadratic trend in yield by increasing tissue amount is 

different when a supplement applied compared to when no supplement is applied (p < 

0.0001). 

  

1.4 What is the power of the test? 
 



Fixed Scenario Elements 

Dependent Variable yield 

Alpha 0.05 

Error Standard Deviation 8.450738 

Total Sample Size 24 

Error Degrees of Freedom 15 

 

Computed Power 

Index Source Test DF Power 

1 undergrad 3 0.995 

2 trtmt 5 >.999 

The power analysis suggests that the power to detect differences between the treatments is Power 

> 0.999. 

 

1.5 Was there any information gained by using the model above vs. a CRD? Please, provide a 

value for efficiency of the current model vs. a CRD. 

To calculate the relative efficiency of the RCBD compared to the CRD design, the following 

formula will be used: 

RERCBD:CRD = 
(𝑑𝑓𝑀𝑆𝐸 𝑅𝐶𝐵𝐷+1)(𝑑𝑓𝑀𝑆𝐸 𝐶𝑅𝐷+3)𝑀𝑆𝐸𝐶𝑅𝐷

(𝑑𝑓𝑀𝑆𝐸 𝐶𝑅𝐷+1)(𝑑𝑓𝑀𝑆𝐸 𝑅𝐶𝐵𝐷+3)𝑀𝑆𝐸𝑅𝐶𝐵𝐷
 

 

Since we don’t know the MSECRD these values must be calculated by using the following 

formula: 

MSECRD = 
𝑑𝑓𝐵𝑙𝑜𝑐𝑘 × 𝑀𝑆𝐵𝑅𝐶𝐵𝐷 + (𝑑𝑓𝑇𝑟𝑒𝑎𝑡𝑚𝑒𝑛𝑡 + 𝑑𝑓𝑒𝑟𝑟𝑜𝑟) 𝑀𝑆𝐸𝑅𝐶𝐵𝐷

𝑑𝑓𝐵𝑙𝑜𝑐𝑘 + 𝑑𝑓𝑇𝑟𝑒𝑎𝑡𝑚𝑒𝑛𝑡 + 𝑑𝑓𝑒𝑟𝑟𝑜𝑟
 

 

Given: dfBlock = 3, dfTreatment = 5, dferror = 15, MSBRCBD =818.923, MSERCBD = 71.415 

 

MSECRD = 
3× 818.923 + (5 +15)71.415

3 + 5 + 15
 = 168.916 

 

Now that we have an estimate for MSECRD we can revisit our first equation: 

 

RERCBD:CRD = 
(𝑑𝑓𝑀𝑆𝐸 𝑅𝐶𝐵𝐷+1)(𝑑𝑓𝑀𝑆𝐸 𝐶𝑅𝐷+3)𝑀𝑆𝐸𝐶𝑅𝐷

(𝑑𝑓𝑀𝑆𝐸 𝐶𝑅𝐷+1)(𝑑𝑓𝑀𝑆𝐸 𝑅𝐶𝐵𝐷+3)𝑀𝑆𝐸𝑅𝐶𝐵𝐷
 

Given: dfMSE RCBD = 15, dfMSE CRD = 18, MSECRD = 168.916, MSERCBD = 71.415 

 

RERCBD:CRD = 
(15+1)(18+3)168.916

(18+1)(15+3)71.415
 = 2.32378 

 



Based on the results above using an RCBD was much more informative than using a CRD. It 

would take 2.32378 ~ 3 replications in the CRD to produce the same amount of information as 

one replication in the RCBD. 

 

 

1.6 Repeat the analysis without accounting for the variation between the undergraduate students 

report the overall ANOVA, the coefficient of determination R
2
, and the ANOVA to test 

differences between treatments. Compare the ANOVA results with the ANOVA results in 1.3. 

Report the differences between: 

Source DF Sum of Squares Mean Square F Value Pr > F 

Model 5 815214.9337 163042.9867 831.85 <.0001 

Error 18 3527.9925 195.9996     

Corrected Total 23 818742.9262       

 

R-Square Coeff Var Root MSE yield Mean 

0.995691 4.029787 13.99999 347.4125 

 

Source DF Type III SS Mean Square F Value Pr > F 

trtmt 5 815214.9338 163042.9868 831.85 <.0001 

 

 

 

a. Degrees of freedom and sums of square of the error. 

The df of the MSE in CRD analysis is larger than the MSE in the RCBD analysis (dfCRD  = 18 > 

dfRCBD = 15). 

The MSE of the CRD analysis is larger than in the MSE of the analysis of the RCBD (MSECRD  = 

195.9996 > MSERCBD = 71.415). 

 

b. The coefficient of determination (R
2
)

 
and what does this value represent? 

The R
2
in the CRD analysis is smaller than the R - squared in the RCBD (R

2 
CRD  = 0.995691 < R

2
 

RCBD = 0.998692). Although in this case it is not much different because the effect of the treatment is 

very large. This value represents how much of the variation in the data is explained by the model. In 

this example the CRD model explains less of the variation than the RCBD model. 

 

c. The F - value for the treatment ANOVA in the CRD and RCBD, Why did the F - 

value change? What changed in the degrees of freedom for error and why? What 

happens to P as F gets smaller and closer to 1? 

The F – value of the ANOVA in the CRD analysis is smaller than the F – value of the ANOVA in 

the RCBD analysis (F-value CRD  = 831.85 < F-value RCBD = 2283.04) and therefore the P value is 

larger (less significant), although in this case both are highly significant. As the F – value approaches 

1 the p – value will become larger and not significant. The Error in the CRD has 18 df and in the 

RCBD 15 df. The 3 df difference are now in the blocking variable “undergrad”.  

 



By using a CRD analysis rather than a RCBD analysis the MSE has increased because the model 

now explained less of the variation (i.e. the variation between undergraduate students is not 

explained by the model). The increase in the MSE has reduced the value of the F – value because the 

F – value is the ratio of variation of the variable in question divided by the variation of the pooled 

variation of all the treatments (in this case the variation not explained by the model or MSE). 

𝐹 =  
𝑠𝑏𝑒𝑡𝑤𝑒𝑒𝑛 𝑡𝑟𝑒𝑎𝑡𝑚𝑒𝑛𝑡𝑠

2

𝑠𝑝𝑜𝑜𝑙𝑒𝑑
2  

 

d. Was there a change in the power of the test? Report the error standard deviation used 

to do the analysis. 

Fixed Scenario Elements 

Method Exact 

Alpha 0.05 

Group Means 100.175 425.275 595.975 102.65 454.85 405.55 

Standard Deviation 13.99999 

Sample Size Per Group 4 

 

Computed Power 

Power 

>.999 

The power of both tests is so high (>0.999) that in this artificial example we cannot see the 

difference between the RCBD and CRD. However, the standard deviation used to do the CRD 

analysis was 13.999, which is larger than the error standard deviation in the RCBD, indicating that 

the power is lower in the CRD. Also OK if Proc GLMPower was used instead of Proc Power. 
 

Question 2 [40 points] 

 

A scientist wants to test if different fermentation processes of cacao beans affect the content of acetic 

acid in the beans after fermentation. 

The different fermentation treatments are: 

1) SmPKt - Small scale fermentation with inoculation of Pichia kluyveri fermentatioin 

microbial strain using tray method 

2) SmKMt - Small scale fermentation with inoculation of Kluyveromyces marxianus 

fermentation microbial strain using tray method 

3) SmCt - Small scale fermentation with no inoculation of microbial strain using tray method 

4) Comt - Commercial scale fermentation using tray method with no inoculation of microbial 

strain 

5) Comh - Commercial scale fermentation using heap method with no inoculation of microbial 

strain 

 



The scientist obtained large amounts of five different varieties of cacao and split each variety into 5 

equal parts that were distributed to five different facilities, each receiving all five varieties. Each 

facility agreed to perform 5 analyses, one for each treatment using different varieties per treatment. 

The scientist gave specific instructions to each facility of what fermentation process (i.e. treatment) 

should be done to each variety. The experimental design is summarized below: 

 

  Fac 1 Fac 2 Fac 3 Fac 4 Fac 5 

Var 1 SmPKt SmKMt SmCt Comt Comh 

Var 2 SmKMt SmCt Comt Comh SmPKt 

Var 3 SmCt Comt Comh SmPKt SmKMt 

Var 4 Comt Comh SmPKt SmKMt SmCt 

Var 5 Comh SmPKt SmKMt SmCt Comt 

 

After the fermentation process was done the scientist received all of the fermented cacao beans. The 

scientist extracted and measured the acetic acid content in each sample that was sent to the facilities. 

The data from this experiment is reported below (Note: You can copy and paste this data into SAS to 

do your analysis): 
Data hw67_2; 

Input Variety $ Facility $ Treatment $ aacid; 

Cards; 

Var1 Fac1 SmPKt 16.06 

Var2 Fac1 SmKMt 9.13 

Var3 Fac1 SmCt 13.01 

Var4 Fac1 Comt 2.21 

Var5 Fac1 Comh 6.59 

Var1 Fac2 SmKMt 16.60 

Var2 Fac2 SmCt 7.26 

Var3 Fac2 Comt 9.35 

Var4 Fac2 Comh 1.10 

Var5 Fac2 SmPKt 10.47 

Var1 Fac3 SmCt 15.22 

Var2 Fac3 Comt 3.82 

Var3 Fac3 Comh 8.13 

Var4 Fac3 SmPKt 6.15 

Var5 Fac3 SmKMt 10.75 

Var1 Fac4 Comt 10.64 

Var2 Fac4 Comh 2.61 

Var3 Fac4 SmPKt 12.15 

Var4 Fac4 SmKMt 5.55 

Var5 Fac4 SmCt 8.88 

Var1 Fac5 Comh 9.75 



Var2 Fac5 SmPKt 5.58 

Var3 Fac5 SmKMt 12.72 

Var4 Fac5 SmCt 4.20 

Var5 Fac5 Comt 5.47 

; 

Proc Print Data = hw67_2; 

title 'check data'; 

 

Proc GLM Data = hw67_2; * notice that the Order statement was not used; 

title 'ANOVA';          * Check the Means results; 

Class Facility Variety Treatment; 

Model aacid = Facility Variety Treatment; 

Output Out = PRhw67_2 r = res p = pred; 

Means Treatment; * Check the Means results to see in what order SAS is 

reading the treatments; 

     * This is necessary to construct the contrast; 

Contrast 'small scale vs commercial'  Treatment -3 -3  2  2  2; 

Contrast 'inoculated vs no-inoculation'   Treatment  0  0 -2  1  1; 

Contrast 'P. kluyveri vs K. marxianus'  Treatment  0  0  0  1 -1; 

Contrast 'heap vs tray'    Treatment  1 -1  0  0  0; 

 

Proc Gplot Data= PRhw67_2; 

title 'Res vs Pred plot'; 

plot res*pred = Treatment; 

Proc Univariate Normal plot Data = PRhw67_2; 

title 'Normality'; 

var res; 

 

Proc GLM Data = hw67_2; 

title 'Levene for treatment'; 

Class Treatment; 

Model aacid = Treatment; 

Means Treatment / HovTest = Levene; 

 

Proc GLM Data = PRhw67_2; * 1- df additivity test btw facility and treatment 

this is the only one requested; 

title 'add. fac. trt.'; 

Class Facility Treatment; 

Model aacid = Facility Treatment; 

Output Out = PRfactrt r = rfactrt p = pfactrt; 

 

Proc GLM Data = PRfactrt; 

Class Facility Treatment; 

Model aacid = Facility Treatment pfactrt*pfactrt; 

 

Proc GLM Data = hw67_2; * 1- df additivity test btw variety and facility. NOT 

REQUESTED IN PROBLEM; 

title 'add. fac. var.'; 

Class Variety Facility; 

Model aacid = Variety Facility; 

Output Out = PRvarfac r = rvarfac p = pvarfac; 

 

Proc GLM Data = PRvarfac; 

title 'add. var. fac.'; 

Class Variety Facility; 

Model aacid = Variety Facility pvarfac*pvarfac; 

Data varfac(drop = Treatment); 



set PRhw67_2; 

 

Proc Print data = varfac; 

Proc Gplot data = varfac; 

plot aacid*pred; 

 

Proc GLM Data = hw67_2; *1- df additivity test btw variety and treatment NOT 

REQUESTED IN PROBLEM; 

title 'add. var. trt.'; 

Class Variety Treatment; 

Model aacid = Variety Treatment; 

Output Out = PRvartrt r = rvartrt p = pvartrt; 

 

Proc GLM Data = PRvartrt; 

Class Variety Treatment; 

Model aacid = Variety Treatment pvartrt*pvartrt; 

 

 

run; quit; 

 
2.1 Describe the design of the experiment using the table in the appendix. 

Design:  Latin Square 

Response Variable:  Acetic acid content 

Experimental Unit:  Sub-batch of cacao beans 

    

Class 

Variable 

Block or 

Treatment 

No. of 

Levels 
Description 

1  Block 5   Cacao variety 

2 Block  5   Fermentation facility 

3  Treatment  5 Bean fermentation method  

    

  Subsamples?  NO   

 

 

2.2 Perform the appropriate ANOVA and report the results and your conclusions. For this design 

and answer the following questions by performing contrast (You do not have to show that 

they are orthogonal): 



 
 

Source DF Type III SS Mean Square F Value Pr > F 

Variety 4 314.1320400 78.5330100 597.80 <.0001 

Facility 4 11.5457200 2.8864300 21.97 <.0001 

Treatment 4 115.1192000 28.7798000 219.07 <.0001 

The ANOVA suggest that there were significant differences in acetic acid content between the 

fermentation method treatments (p < 0.0001). There was a significant effect of variety on acetic acid 

content (p < 0.0001). There was a significant effect of facility on acetic acid content (p < 0.0001). 

 

a. Is there a difference in acetic acid content between fermentation treatments which 

were done at small scale compared to large scale? 

Contrast DF Contrast SS Mean Square F Value Pr > F 

small scale vs commercial 1 109.9960167 109.9960167 837.30 <.0001 

The 1 – df test suggest that there was a significant difference between the treatments that were done 

at a small scale (i.e. experimental scale) and the treatments that were done at a commercial scale (p < 

0.0001). 

 

b. Within the small scale experiments is there a difference in acetic acid content 

between the inoculated treatments and the non-inoculated treatment? 

Contrast DF Contrast SS Mean Square F Value Pr > F 

inoculated vs no-inoculation 1 2.1440133 2.1440133 16.32 0.0016 

The 1 – df test suggest that within the treatments that were done at a small scale, there was a 

significant difference between the treatments that were inoculated with a fermenting microbial strain 

and the treatments that were not inoculated (p = 0.0016). 

c. Within the small scale experiments is there a difference in acetic acid content 

between the P. kluyveri strain and the K. marxianus strain? 

Contrast DF Contrast SS Mean Square F Value Pr > F 



Contrast DF Contrast SS Mean Square F Value Pr > F 

P. kluyveri vs K. marxianus 1 1.8835600 1.8835600 14.34 0.0026 

Within the treatments that were inoculated with the fermenting microbial strain, there were 

significant differences in acetic acid content between inoculating with P. kluyveri and inoculating 

with K. marxianus (P = 0.0026). 

 

d. Between the large scale experiments is there a difference in acetic acid content 

between the heap and tray fermentation processes 

Contrast DF Contrast SS Mean Square F Value Pr > F 

heap vs tray 1 1.0956100 1.0956100 8.34 0.0136 

Within the treatments that were done at a large scale, there were significant differences in acetic acid 

between treatments using the heap method compared to the tray method (p = 0.0136). 

 

2.3 Test the assumptions of normality of residuals. Test for homogeneity of variances. The 

homogeneity of variances test should be done for the treatment effect only. 

Tests for Normality 

Test Statistic p Value 

Shapiro-Wilk W 0.97367 Pr < W 0.7385 

The Shapiro – Wilk test for normality is not significant; therefore we conclude that the data follows 

a normal distribution (p = 0.7385). 

Levene's Test for Homogeneity of aacid Variance 

ANOVA of Squared Deviations from Group Means 

Source DF Sum of Squares Mean Square F Value Pr > F 

Treatment 4 120.7 30.1811 0.20 0.9355 

Error 20 3021.9 151.1   
 

The Levene test for homogeneity of variances is not significant; therefore, we conclude that the 

variances between treatments are not significantly different (p = 0.9355). 

 

2.4 Perform the Tukey 1- df test to test if there are multiplicative effects between fermentation 

facilities and treatments. 

Source DF Type III SS Mean Square F Value Pr > F 

Facility 4 27.59304055 6.89826014 0.37 0.8232 

Treatment 4 29.26308965 7.31577241 0.40 0.8075 

pfactrt*pfactrt 1 39.45771561 39.45771561 2.14 0.1639 

The Tukey 1 – df test suggest that there are not significant multiplicative effects between 

fermentation facilities and treatments (p = 0.1639). 

 

2.5   Estimate the relative efficiency of this design compared with a RCBD assuming that:  



a. Fermentation facilities are not included as a classification variable (RCBD with 

varieties as blocks)  

To calculate the relative efficiency of the Latin Square design compared to the RCBD where 

facilities are removed as a blocking variable first it is necessary to estimate the MSE of having 

an RCBD where we remove facilities as a blocking variable. This could be done by using the 

flowing equation: 

 

MSE RCBD if facilities is removed =  
𝑑𝑓𝑓𝑎𝑐𝑖𝑙𝑖𝑡𝑖𝑒𝑠 𝐿𝑆×𝑀𝑆𝐹𝑎𝑐𝑖𝑙𝑖𝑡𝑖𝑒𝑠𝐿𝑆+(𝑑𝑓𝑇𝑟𝑒𝑎𝑡𝑚𝑒𝑛𝑡 𝐿𝑆+𝑑𝑓𝑒𝑟𝑟𝑜𝑟 𝐿𝑆)×𝑀𝑆𝐸𝐿𝑆

𝑑𝑓𝑓𝑎𝑐𝑖𝑙𝑖𝑡𝑖𝑒𝑠 𝐿𝑆+𝑑𝑓𝑇𝑟𝑒𝑎𝑡𝑚𝑒𝑛𝑡 𝐿𝑆+𝑑𝑓𝑒𝑟𝑟𝑜𝑟 𝐿𝑆
 

 

Where: df facilities LS = 4, df Treatment LS = 4, df error LS = 12, MSFacilities LS = 2.88643, MSE LS = 

0.13137. 

MSE RCBD if facilities is removed =  
4 × 2.88643+(4 + 12)×0.13137

4 + 4 + 12
 =0.68238 

 
Now we can calculate the relative efficiency of the Latin Square design with a RCBD where the 

facility was removed as a blocking variable. To do this we use the following equation: 

 

RE LS : RCBD facilities removed = 
(𝑑𝑓𝑀𝑆𝐸 𝐿𝑆+1)(𝑑𝑓𝑀𝑆𝐸 𝑅𝐶𝐵𝐷+3)𝑀𝑆𝐸𝑅𝐶𝐵𝐷 𝑓𝑎𝑐𝑖𝑙𝑖𝑡𝑒𝑠 𝑟𝑒𝑚𝑜𝑣𝑒𝑑

(𝑑𝑓𝑀𝑆𝐸 𝑅𝐶𝐵𝐷+1)(𝑑𝑓𝑀𝑆𝐸 𝐿𝑆+3)𝑀𝑆𝐸𝐿𝑆
 

 

Where: df MSE LS = 12, df MSE RCBD = 16, MSE RCBD facilities removed = 0.68238, MSE LS = 0.13137 

 

RE LS : RCBD facilities removed = 
(12+1)(16+3)0.68238

(16+1)(12+3)0.13137
 = 5.03139 

The relative efficiency of the Latin square design where we included facilities as a blocking 

variable was more efficient than an RCBD where we remove facilities as a blocking variable. 

Based on the result above it would take ~5 replications in an RCBD without blocking for 

facilities to produce the same information as 1 replication in a LS design. 

 

b. Varieties are not included as a classification variable 

To calculate the relative efficiency of the Latin Square design compared to the RCBD where 

varieties are removed as a blocking variable it is necessary to first estimate the MSE of the 

RCBD where varieties were removed as a blocking variable: 

 

MSE RCBD if varieties is removed =  
𝑑𝑓𝑣𝑎𝑟𝑖𝑒𝑡𝑖𝑒𝑠 𝐿𝑆×𝑀𝑆𝑉𝑎𝑟𝑖𝑒𝑡𝑖𝑒𝑠𝐿𝑆+(𝑑𝑓𝑇𝑟𝑒𝑎𝑡𝑚𝑒𝑛𝑡 𝐿𝑆+𝑑𝑓𝑒𝑟𝑟𝑜𝑟 𝐿𝑆)×𝑀𝑆𝐸𝐿𝑆

𝑑𝑓𝑣𝑎𝑟𝑖𝑒𝑡𝑖𝑒𝑠 𝐿𝑆+𝑑𝑓𝑇𝑟𝑒𝑎𝑡𝑚𝑒𝑛𝑡 𝐿𝑆+𝑑𝑓𝑒𝑟𝑟𝑜𝑟 𝐿𝑆
 

 

Where: df varieties LS = 4, df Treatment LS = 4, df error LS = 12, MSVarieties LS = 78.53301, MSE LS = 

0.13137. 

 

MSE RCBD if varieties is removed =  
4 ×78.5330100+ (4 +12)×0.13137

4+4+12
 = 15.8117 

 
Now we can calculate the RE between using the LS and an RCBD where varieties were removed 

as a blocking variable using the same equation as in problem 2.5a with a change in the MSE 

RCBD: 



 

RE LS : RCBD facilities removed = 
(𝑑𝑓𝑀𝑆𝐸 𝐿𝑆+1)(𝑑𝑓𝑀𝑆𝐸 𝑅𝐶𝐵𝐷+3)𝑀𝑆𝐸𝑅𝐶𝐵𝐷 𝑣𝑎𝑟𝑖𝑒𝑡𝑖𝑒𝑠 𝑟𝑒𝑚𝑜𝑣𝑒𝑑

(𝑑𝑓𝑀𝑆𝐸 𝑅𝐶𝐵𝐷+1)(𝑑𝑓𝑀𝑆𝐸 𝐿𝑆+3)𝑀𝑆𝐸𝐿𝑆
 

 

Where: df MSE LS = 12, df MSE RCBD = 16, MSE RCBD facilities removed = 15.8117, MSE LS = 0.13137 

 

RE LS : RCBD facilities removed = 
(12+1)(16+3)15.8117

(16+1)(12+3)0.13137
 = 116.584 

 

The results above demonstrate that using the LS was much more efficient than using an RCBD 

without variety as a blocking variable. It would take 116.6 ~ 117 replications in an RCBD where 

varieties were not included as a blocking variable to produce the same information as 1 

replication in a LS design. 

 

 
 

Question 3 [20 points] 

 

A lettuce grower received lettuce seed from 10 new lettuce varieties.  The grower conducted a field 

experiment to measure the average lettuce weight for these new lettuce varieties. 

 

To conduct the experiment the grower planted each lettuce variety in 20 ft long plots. The 

grower suspects of uneven soil fertility in the field; therefore, to perform his experiment he 

divided his field into 4 blocks of even size. Each block had 20 plots laid out. The grower planted 

two replications of each variety twice in each block. The order in which individual replications 

of each cultivar were planted within the block was completely random. When the lettuce reached 

harvest maturity the grower harvested 24 lettuce plants from individual plots and weighed them 

in bulk (the grower did not measure each plot more than once). The grower then divided the total 

weight of the lettuce heads by the number of heads weighed to calculate the average lettuce 

weight (grams) for the given plot. 

 

The data for this experiment is summarized below: 

 
Block 1 Block 2 Block 3 Block 4 

Var1 223.7 218.2 339.2 278.7 264.6 303.3 399.6 365.4 

Var2 294.4 312.5 325.6 310.5 419.8 367.9 468.7 420.8 

Var3 342.7 336.8 357.7 410.4 415.0 418.3 477.5 470.2 

Var4 416.2 402.1 477.0 441.2 511.9 475.6 614.5 554.9 

Var5 392.3 403.7 467.9 547.0 579.4 557.4 607.1 618.4 

Var6 576.2 546.7 566.8 579.7 601.3 617.3 633.7 658.9 

Var7 663.6 676.0 706.4 730.8 789.2 812.8 771.5 824.2 

Var8 724.7 719.5 822.5 792.8 897.4 859.5 963.5 874.6 

Var9 956.9 871.5 983.7 1108.7 1060.7 1035.2 1073.7 1169.5 

Var10 1191.9 1232.7 1199.2 1268.5 1386.4 1226.1 1284.7 1280.5 
 

(NOTE: The table above does not represent the physical layout of the field) 



 
Data hw67_3; 

Input Block $ Variety $ plantfreshwt @@; 

Cards; 

Block1 Var1 223.7 Block2 Var1 339.2 Block3 Var1 264.6 Block4 Var1 399.6 

Block1 Var2 294.4 Block2 Var2 325.6 Block3 Var2 419.8 Block4 Var2 468.7 

Block1 Var3 342.7 Block2 Var3 357.7 Block3 Var3 415.0 Block4 Var3 477.5 

Block1 Var4 416.2 Block2 Var4 477.0 Block3 Var4 511.9 Block4 Var4 614.5 

Block1 Var5 392.3 Block2 Var5 467.9 Block3 Var5 579.4 Block4 Var5 607.1 

Block1 Var6 576.2 Block2 Var6 566.8 Block3 Var6 601.3 Block4 Var6 633.7 

Block1 Var7 663.6 Block2 Var7 706.4 Block3 Var7 789.2 Block4 Var7 771.5 

Block1 Var8 724.7 Block2 Var8 822.5 Block3 Var8 897.4 Block4 Var8 963.5 

Block1 Var9 956.9 Block2 Var9 983.7 Block3 Var9 1060.7 Block4 Var9 1073.7 

Block1 Var10 1191.9 Block2 Var10 1199.2 Block3 Var10 1386.4 Block4 Var10 1284.7 

Block1 Var1 218.2 Block2 Var1 278.7 Block3 Var1 303.3 Block4 Var1 365.4 

Block1 Var2 312.5 Block2 Var2 310.5 Block3 Var2 367.9 Block4 Var2 420.8 

Block1 Var3 336.8 Block2 Var3 410.4 Block3 Var3 418.3 Block4 Var3 470.2 

Block1 Var4 402.1 Block2 Var4 441.2 Block3 Var4 475.6 Block4 Var4 554.9 

Block1 Var5 403.7 Block2 Var5 547.0 Block3 Var5 557.4 Block4 Var5 618.4 

Block1 Var6 546.7 Block2 Var6 579.7 Block3 Var6 617.3 Block4 Var6 658.9 

Block1 Var7 676.0 Block2 Var7 730.8 Block3 Var7 812.8 Block4 Var7 824.2 

Block1 Var8 719.5 Block2 Var8 792.8 Block3 Var8 859.5 Block4 Var8 874.6 

Block1 Var9 871.5 Block2 Var9 1108.7 Block3 Var9 1035.2 Block4 Var9 1169.5 

Block1 Var10 1232.7 Block2 Var10 1268.5 Block3 Var10 1226.1 Block4 Var10 1280.5 

; 

Proc Print Data = hw67_3; 

title 'Print data'; 

Proc GLM Data = hw67_3; 

title 'ANOVA exploratory'; * this model is used to test if there are Block by variety 

interactions; 

Class Block Variety; 

Model plantfreshwt = Block Variety Block*Variety; 

Proc GLM Data = hw67_3; 

Title 'ANOVA'; 

Class Block Variety; 

Model plantfreshwt = Block Variety; 

Output Out = PRhw67_3 r = res p = pred; 

Means Variety / Tukey; 

Proc Univariate Normal Plot Data = PRhw67_3; 

title 'Normality'; 

var res; 

Proc GLM Data = hw67_3; 

title 'Levene'; 

Class Variety; 

Model plantfreshwt = Variety; 

Means Variety / HovTest = Levene; 

Proc Gplot Data = PRhw67_3; 

plot res*pred = Variety; 

run; quit; 

 

3.1 Describe in detail the design of this experiment using the table in the appendix. 

Design:  RCBD with two replications per block 

Response Variable:  Average plant weight (g) 

Experimental Unit:  Field plot 

    

Class 

Variable 

Block or 

Treatment 

No. of 

Levels 
Description 

1 Block  4   Field blocks of equal size 

2 Treatment  10   Lettuce varieties 



↓       

n       

    

  Subsamples? NO   

  

 

3.2  Test the assumptions of normality of residuals, homogeneity of variances between treatments and 

test for additivity of the block and treatment effects. 

Normality of Residuals: 

Tests for Normality 

Test Statistic p Value 

Shapiro-Wilk W 0.981602 Pr < W 0.3046 

The Shapiro – Wilk test suggest that the data fits a normal distribution (p = 0.3046). 

 

Homogeneity of Variances: 

Levene's Test for Homogeneity of plantfreshwt Variance 

ANOVA of Squared Deviations from Group Means 

Source DF Sum of Squares Mean Square F Value Pr > F 

Variety 9 2.9449E8 32720681 1.23 0.2890 

Error 70 1.8563E9 26518894     

The Levene test suggest that the variances between varieties are homogeneous (p = 0.2890). 

 

Multiplicity of block and treatment effects: 

Source DF Type III SS Mean Square F Value Pr > F 

Block 3 250008.088 83336.029 60.99 <.0001 

Variety 9 6960685.423 773409.491 566.01 <.0001 

Block*Variety 27 41170.667 1524.840 1.12 0.3696 

Since there are two replications per block it is possible to run an exploratory model to test for block by 

variety interactions, which is the multiplicity of block and treatment effects. The results of the exploratory 

model suggest that there are no significant block by variety interactions (p = 0.3696). 

 

3.3    Run the correct ANOVA model to test if there are significant differences in average lettuce weight 

between the 10 varieties and conduct a Tukey pairwise-comparison test to determine which varieties have 

significantly higher average lettuce weights than others and report the minimum significant difference 

used to establish a significant difference in the Tukey test. 

Source DF Type III SS Mean Square F Value Pr > F 

Block 3 250008.088 83336.029 58.27 <.0001 

Variety 9 6960685.423 773409.491 540.75 <.0001 

Since the block by variety interactions were not significant the block by variety interaction can be 

excluded from the model to get a more sensitive ANOVA (i.e. an ANOVA that is better at detecting 



differences). The results of the ANOVA suggest that there were significant differences in average plant 

weight (g) between the varieties tested (p < 0.0001). 

 

Alpha 0.05 

Error Degrees of Freedom 67 

Error Mean Square 1430.258 

Critical Value of Studentized Range 4.62808 

Minimum Significant Difference 61.882 

 

Means with the same letter 

are not significantly different. 

Tukey Grouping Mean N Variety 

A 1258.75 8 Var10 

B 1032.49 8 Var9 

C 831.81 8 Var8 

D 746.81 8 Var7 

E 597.58 8 Var6 

F 521.65 8 Var5 

F 486.68 8 Var4 

G 403.58 8 Var3 

G 365.03 8 Var2 

H 299.09 8 Var1 

The least significant difference in the Tukey pairwise – comparison test is 61.882 g. The Tukey 

pairwise – comparison test demonstrates that Var 1 has the lowest average plant weight (g) and is 

significantly different from all other varieties. Varieties 10, 9, 8, and 7 are the top 4 varieties 

with the highest average plant weight in that order. Varieties 5 and 4 are not significantly 

different in average plant weight and are grouped together. Varieties 3 and 2 are not significantly 

different in average plant weight and are grouped together. 
  

 

Appendix 

 

When you are asked to "describe in detail the design of this experiment," please do so by 

completing the following template: 

 

Design:   

Response Variable:   



Experimental Unit:   

    

Class 

Variable 

Block or 

Treatment 

No. of 

Levels 
Description 

1       

2       

↓       

n       

    

  Subsamples? YES / NO   

 
NOTICE:  There is a new column in the above table ("Block or Treatment").  Now, for each class 

variable in your model, you need to indicate if it is a "Block" variable or a "Treatment" variable. 

 

 

 


